# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=hls-splice-plugin
pkgname=haskell-hls-splice-plugin
pkgver=1.0.3.0
pkgrel=99
pkgdesc="HLS Plugin to expand TemplateHaskell Splices and QuasiQuotes"
url="https://hackage.haskell.org/package/hls-splice-plugin"
license=("Apache")
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-aeson' 'haskell-dlist' 'haskell-extra' 'haskell-foldl' 'haskell-ghc'
         'haskell-ghc-exactprint' 'haskell-ghcide' 'haskell-hls-plugin-api'
         'haskell-hls-refactor-plugin' 'haskell-lens' 'haskell-lsp' 'haskell-retrie' 'haskell-syb'
         'haskell-unliftio-core' 'haskell-unordered-containers')
makedepends=('ghc' 'uusi' 'haskell-hls-test-utils')
checkdepends=('git')
source=("https://hackage.haskell.org/packages/archive/$_hkgname/$pkgver/$_hkgname-$pkgver.tar.gz")
sha512sums=('c021d8b78c707277e64f44f71abda4f6fac866487d0104f1062dc13c151ee19aca290d2d96258bdf7b51aa3626513e052dacd27826f8bda74d7714784cbc9563')

prepare() {
  cd $_hkgname-$pkgver
  gen-setup
  uusi -u ghcide -u hls-plugin-api -u hls-test-utils $_hkgname.cabal
}

build() {
  cd $_hkgname-$pkgver

  runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
    --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
    --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm \
    --ghc-option=-optl-Wl\,-z\,relro\,-z\,now \
    --ghc-option='-pie'

  runhaskell Setup build $MAKEFLAGS
  runhaskell Setup register --gen-script
  runhaskell Setup unregister --gen-script
  sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
  sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
  cd $_hkgname-$pkgver
  runhaskell Setup test --show-details=direct
}

package() {
  cd $_hkgname-$pkgver

  install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
  install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
  runhaskell Setup copy --destdir="$pkgdir"
  rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
